

Solutions to Selected Exercises

To derive the maximum benefit from the exercises, make an honest effort to create the Maple code before consulting the solution offered here.

▼ Part I. The Maple Worksheet

▼ Section 1. Execution Groups: Input/Output

2.* Simplify the square root of 19,220 by entering and executing

```
sqrt(19220);
```

Then obtain the complete integer factorization of 19220 by entering and executing

```
ifactor(19220);
```

Read **ifactor** as "integer factors of".

```
> sqrt(19220);
```

$$62\sqrt{5} \quad (1.1.1)$$

```
> ifactor(19220);
```

$$(2)^2 (5) (31)^2 \quad (1.1.2)$$

3.* Obtain an integral of the expression

$$e^{3x} \cos(2x) \sin(4x).$$

Enter it into the **int** procedure as follows

```
int( exp(3*x)*cos(2*x)*sin(4*x), x);
```

```
> int( exp(3*x)*cos(2*x)*sin(4*x), x);
```

$$-\frac{1}{15} e^{3x} \cos(6x) + \frac{1}{30} e^{3x} \sin(6x) + \frac{3}{13} e^{3x} \cos(2x) - \frac{9}{26} e^{3x} \sin(2x) \quad (1.1.3)$$

$$+ \frac{2}{13} e^{3x} (3 \sin(2x) - 2 \cos(2x))$$

▼ Section 2. Entering blocks of text

18.* Save your worksheet (**Command-S**) then print it.

See the pdf file entitled My_2nd_wrksht.pdf.

▼ Part II. Calculations and Calculus with Maple

▼ Section 1. Getting Started: Maple as a Calculator

6. Use the sequence operator, \$, to make the following sequences.

a. 2, 4, 6, 8, 10, 12

b. 20, 40, 60, ... , 260 Hint. $20 + 20*k$ \$ k=

c. Now use the sequence procedure, **seq**, to make th sequence in part b.

7.* Add the numbers in the sequence in 6 c. Hint. If you just did 6 c, then execute **add(k,k=%)**.

```
> seq(k, k = 20..260, 20);  
20, 40, 60, 80, 100, 120, 140, 160, 180, 200, 220, 240, 260 (2.1.1)
```

```
> add(k,k=%);  
1820 (2.1.2)
```

10. Use **seq** to obtain the sequence of the cubes of the numbers in the list [2, 5, 6, 9, 12, 44]. Add the numbers in the sequence of cubes and then obtain the prime factorization of the sum.

What do you notice about the prime factorization? (Hint. The integer 87990 is called "square free".)

11.* Obtain the prime factorization of the product of the integers in the sequence of cubes described in Exercise 10.

```
> seq(k^3, k=[2,5,6,9,12,44]);  
8, 125, 216, 729, 1728, 85184 (2.1.3)
```

```
> mul(k,k=%);  
ifactor(%);  
23178378313728000  
(2)^18 (3)^12 (5)^3 (11)^3 (2.1.4)
```

12. Use the **factor** procedure to factor the following polynomial expressions.

a. $x^3 - x^2 + x - 1$

b. $x^7 - x^6 + x^5 - \dots - 1$ Hint. Enter this as **add(-(-x)^k,k=0..7)** then **factor** (%).

15.* Use solve to obtain the zeros of the polynomials in Exercise 12.

```
> x^3 - x^2 + x - 1;  
solve(%);  
x^3 - x^2 + x - 1  
1, I, -I (2.1.5)
```

```
> add(-(-x)^k,k=0..7);  
solve(%);  
-1 + x - x^2 + x^3 - x^4 + x^5 - x^6 + x^7  
1, I, -I, 1/2*sqrt(2) + 1/2*I*sqrt(2), -1/2*sqrt(2) + 1/2*I*sqrt(2), -1/2*sqrt(2) - 1/2*I*sqrt(2), 1/2*sqrt(2) - 1/2*I*sqrt(2) (2.1.6)
```

16.* Add the zeros of the polynomial in 12 a and the zeros of the polynomial in 12 b.

```
> solve(x^3 - x^2 + x - 1);
add(k,k=%);
```

$$1, I, -I$$

$$1$$
(2.1.7)

```
> solve(-add((-x)^k,k=0..7));
add(k,k=%);
```

$$1, I, -I, \frac{1}{2}\sqrt{2} + \frac{1}{2}I\sqrt{2}, -\frac{1}{2}\sqrt{2} + \frac{1}{2}I\sqrt{2}, -\frac{1}{2}\sqrt{2} - \frac{1}{2}I\sqrt{2}, \frac{1}{2}\sqrt{2} - \frac{1}{2}I\sqrt{2}$$

$$1$$
(2.1.8)

▼ Section 2. Symbolics: Equations and Assignments

6.* Plot the expression

$$x^3 - x^2 + 0.05x + \cos(x) - 0.7$$

with the entries

```
y := x^3 - x^2 + 0.05*x + cos(x) - 0.7;
plot(y, x=-2..2, -2..2);
```


Enter and execute **fsolve(y=0,x)** to see which zero **fsolve** finds.

Using the graph as a guide, obtain an approximation to the largest positive zero using

```
fsolve(y=0, {x=a..b}).
```

```
> y := x^3 - x^2 + 0.05*x + cos(x) - 0.7;
plot(y, x=-2..2, -2..2);
```

$$y := x^3 - x^2 + 0.05x + \cos(x) - 0.7$$



```
> fsolve(y=0,x);
```

$$0.6180709010$$
(2.2.1)

```
> fsolve(y=0, {x=1..2});
```

$$\{x = 1.189778784\}$$
(2.2.2)

9. Graph the function

$$y = \cos(x^2) - \sin(2x)$$

over the interval from $x = 0$ to $x = 2$. Define yp as its derivative and find the zero of yp near $x = 1.5$. Name it $xmin$. Then find the minimum value of y over this interval using `eval(y,x=xmin)`. Name it $ymin$.

Use the following entry to plot the graph and the low point.

```
plot( [y, [ [xmin,ymin] ] ], x=0..2, style=[line,point]);
```


10.* Find the area of the region between the graph in Exercise 9 and the x-axis.

Hint: Numerically integrate the absolute value of y from $x = 0$ to $x = 2$ via the entry

```
evalf( Int( abs(y), x=0..2) );
```

```
> y := cos(x^2) - sin(2*x);
plot(y, x=0..2);
```

$$y := \cos(x^2) - \sin(2x)$$



```
> evalf(Int(abs(y), x=0..2));
```

0.9147700913

(2.2.3)

▼ Section 3. Functions as Transformations

5.* An animation.

The tangent line plot in Exercise 3 can easily be animated as follows.

First define the function $T(a,x)$ whose value at (a,x) is the formula for the tangent line to the graph of f at $(a,f(a))$.

$$T := (a,x) \rightarrow f(a) + g(a)*(x - a);$$

Then use

```
plots[animate](plot, [ [f(x), [t,T(a,t),t=a-0.5..a+0.5] ],
                       x=-1..1,-1..2,color=[red,blue]],
a = -1..1);
```

Once the plot appears, click on it with the mouse and the context bar becomes a row of video controls; enjoy.


Read the Help page for `plots[animate]`.

```
> f := x -> cos(x) - x*tan(x): g := D(f):
T := (a,x) -> f(a) + g(a)*(x - a);
```

$$T := (a,x) \rightarrow f(a) + g(a)(x - a)$$

(2.3.1)

```
> plots[animate](plot, [ [f(x), [t,T(a,t),t=a-0.5..a+0.5] ],
                        x=-1..1, -1..2, color=[red,blue] ],
a = -1..1);
```



8.* Using **unapply** and **Matrix**.

Convert the second solution in Exercise 6 into a function f using

```
f := unapply(rhs(soln), x);
```

Evaluate the solution at $x = 0, 0.2, 0.4, \dots, 1.0$ to 4 digits as follows

```
evalf[4]( f(0.2*k) $ k=0..5 );
```

Make a table of function values with the following Matrix entry

```
Matrix( [ [x, 0.2*k $ k=0..5] , [ 'f(x)', % ] ] );
```

```
> restart: deqn := diff(y(x),x) + x*y(x) = x:
soln := dsolve( {deqn,y(0)=0} ):
f := unapply(rhs(soln),x);
```

$$f := x \rightarrow 1 - e^{-\frac{1}{2}x^2} \quad (2.3.2)$$

```
> evalf[4](f(0.2*k) $ k=0..5);
```

$$0., 0.0198, 0.0769, 0.1647, 0.2739, 0.3935 \quad (2.3.3)$$

```
> Matrix( [ [x, 0.2*k $ k=0..5] , [ 'f(x)', % ] ] );
```

$$\begin{bmatrix} x & 0. & 0.2 & 0.4 & 0.6 & 0.8 & 1.0 \\ f(x) & 0. & 0.0198 & 0.0769 & 0.1647 & 0.2739 & 0.3935 \end{bmatrix} \quad (2.3.4)$$

▼ Part III. First Order Ordinary Differential Equations

▼ Section 1. Entering, Solving, Plotting

7.* Enter the differential equation $y' + y = \cos(t)$ with the name DE. Obtain the solution satisfying the generic initial condition $y(t_0) = y_0$. Name it soln.

```
> DE := diff(y(x),x) + y(x) = cos(x);
soln := dsolve( {DE, y(x0)=y0} );
```

$$DE := \frac{d}{dx} y(x) + y(x) = \cos(x)$$

$$\text{soln} := y(x) = \frac{1}{2} \cos(x) + \frac{1}{2} \sin(x) - \frac{e^{-x} \left(-y_0 + \frac{1}{2} \cos(x_0) + \frac{1}{2} \sin(x_0) \right)}{e^{-x_0}} \quad (3.1.1)$$

a. Use the entry **eval(soln, x=x0)** to verify that soln satisfies the initial condition.

> **eval(soln, x=x0);**

$$y(x_0) = y_0 \quad (3.1.2)$$


b. Use **unapply** to make **rhs(soln)** into a function of x, x0, and y0 named phi (see pages 36-37 in Part III, Section 1 of the manual).

> **phi := unapply(rhs(soln), x, x0, y0);**

$$\phi := (x, x_0, y_0) \rightarrow \frac{1}{2} \cos(x) + \frac{1}{2} \sin(x) - \frac{e^{-x} \left(-y_0 + \frac{1}{2} \cos(x_0) + \frac{1}{2} \sin(x_0) \right)}{e^{-x_0}} \quad (3.1.3)$$


c. Use phi to plot some solutions starting at points evenly spaced on the y axis.

> **plot([[x, phi(x, 0, y0), x=0..10] \$ y0=-2..2]);**




d. Use phi to plot some solutions starting at points evenly spaced on the x axis.

> **plot([[x, phi(x, x0, 0), x=x0..10] \$ x0=-2..2]);**




e. Use phi to plot the solutions starting at points evenly spaced around the unit circle in the style of the two plots on page 35 of the manual. That is, one picture runs time forward, another runs time backward.

> **Points := [cos(k*Pi/6), sin(k*Pi/6)] \$ k=0..11;**
plot([[x, phi(x, cos(k*Pi/6), sin(k*Pi/6)), x=cos(k*Pi/6)..2]
\$ k=0..11, [Points]], style=[line\$12, point],
color=[red\$12, black], view=[-2..2, -2..2]);



```
> plot( [ [x, phi(x, cos(k*Pi/6), sin(k*Pi/6)), x=-2..cos(k*Pi/6)]
$ k=0..11, [ Points ] ], style=[line$12, point],
color=[red$12, black], view=[-2..2, -2..2] );
```



▼ Section 2. Working with Solutions: Modeling

3.* Suppose you borrow \$12,000 to buy a car. The loan is to be paid in 60 equal monthly installments at an interest rate of 5% per year.

a. Assume the payments are actually made continuously at whatever rate is needed to pay off the loan in 60 months. Determine the continuous rate per month that would be required.

```
> DE := diff(P(t), t) = 0.05 * P(t) - Q;
soln := dsolve( {DE, P(0)=12000} );
```

$$DE := \frac{d}{dt} P(t) = 0.05 P(t) - Q$$

$$soln := P(t) = 20 Q + e^{\frac{1}{20} t} (12000 - 20 Q) \quad (3.2.1)$$

```
> eqn := subs(P(t)=0, t=5, soln);
Q := solve(eqn, Q);
```

$$eqn := 0 = 20 Q + e^{\frac{1}{4}} (12000 - 20 Q)$$

$$Q := \frac{600 e^{\frac{1}{4}}}{-1 + e^{\frac{1}{4}}} \quad (3.2.2)$$

> **Monthly_Rate = evalf(Q/12)*dollars;**

$$\text{Monthly_Rate} = 226.0405830 \text{ dollars} \quad (3.2.3)$$

b. Compare the answer to part a to the answer if 60 equal monthly payments are made at a constant annual interest rate of 5% applied to the outstanding balance. In other words, the first payment, due one month after the loan is made, would be

$$12000 \left(e^{\frac{0.05}{12}} - 1 \right) + P_1$$

where $P[1]$ is the amount that is put towards reducing the principal (the \$12000) in the first month. The second payment is

$$(12000 - P_1) \left(e^{\frac{0.05}{12}} - 1 \right) + P_2$$

where $P[2]$ is the amount put towards reducing the principal in the second month. The n th payment is

$$\left(12000 - \left(\sum_{k=1}^{n-1} P_k \right) \right) \left(e^{\frac{0.05}{12}} - 1 \right) + P_n.$$

Hint for b. The second payment is supposed to equal to the first. Therefore, the following equation must be satisfied.

```
> eqn := 12000*(exp(0.05/12)-1)+P[1] =
      (12000 - P[1])*exp(0.05/12)-1 + P[2];
      eqn := 50.10431 + P1 = 50.10430800 - 0.004175359 P1 + P2
      (3.2.4)
```

This determines $P[2]$ in terms of $P[1]$.

```
> P[2] := solve(eqn,P[2]);
      P2 := 0.000002000000000 + 1.004175359 P1
      (3.2.5)
```

Create a **for..do** loop that calculates $P[3]$, $P[4]$, ... , $P[60]$ in terms of $P[1]$, then find $P[1]$ using the fact that

$$\sum_{k=1}^{60} P_k = 12000$$

```
> for n from 3 to 60
  do
    eqn :=
      (12000-add(P[k],k=1..n-2))*exp(0.05/12) - 1 + P[n-1] =
      (12000-add(P[k],k=1..n-1))*exp(0.05/12) - 1 + P[n]:
    P[n] := solve(eqn,P[n]):
  end do:
  unassign('n');
> add(P[k],k=1..60) = 12000:
  P[1] := solve(%,P[1]);
      P1 := 176.4078443
      (3.2.6)
```

```

> Check;
for n from 1 to 60 by 19
do
  payment, n, equals, (12000-add(P[k],k=1..n-1))*
                    (exp(0.05/12) - 1) + P[n];
end do;
unassign('n');

```

Check

```

payment, 1, equals, 226.5121523
payment, 20, equals, 226.5121541
payment, 39, equals, 226.5121536
payment, 58, equals, 226.5121537

```

(3.2.7)

In the first payment, $P[1] = \$176.41$, and the interest payment is about \$50.10. The 60th payment is almost all towards the principal.

```

> Last_Principal_Payment=P[60];
Last_Principal_Payment=225.5703165

```

(3.2.8)

▼ Section 3. Slope Fields: DEplot


2.* Use DEplot to make a nice-looking slope field for the autonomous equation $y' = \sin(y)$. (Use the window $t=-6..6, y=-6..6$.)

```

> with(DEtools):
> DE := diff(y(x),x) = sin(y(x));
DEplot( DE, y(x), x=-6..6, y=-6..6, arrows=line,
        dirgrid=[17,17] );

```

$$DE := \frac{d}{dx} y(x) = \sin(y(x))$$




a. Put some solution curves into the plot by adding

```
{ [y(0)=k] $ k=-6..6 }, linecolor=blue.
```

Comment on the relationship between one curve and the next.

```
> DEplot( DE, y(x), x=-6..6, y=-6..6, arrows=line,
```

```
dirgrid=[17,17], { [y(0)=k] $ k=-6..6 },
linecolor=blue);
```



Each curve is a horizontal translate of the other.

b. Obtain the general solution using `dsolve`.

```
> dsolve( DE );
```

$$y(x) = \arctan \left(\frac{2 e^x _CI}{1 + e^{2x} _CI^2}, \frac{-e^{2x} _CI^2 + 1}{1 + e^{2x} _CI^2} \right) \quad (3.3.1)$$

c. Use `dsolve` to obtain the solution satisfying the initial condition $y(0) = 1$. What is the value of this solution when $x = 1$? Get the exact value and an approximation. If the solution is called "soln" use

```
eval(soln,x=1); evalf(%);
```

```
> soln := dsolve( {DE,y(0)=1} );
```

$$\text{soln} := y(x) = \arctan \left(\frac{2 e^x \sin(1) (-1 + \cos(1))}{-\sin(1)^2 - 2 (e^x)^2 + 2 (e^x)^2 \cos(1) + (e^x)^2 \sin(1)^2}, \right. \\ \left. \frac{-2 (e^x)^2 \cos(1) + \sin(1)^2 + (e^x)^2 \sin(1)^2 - 2 (e^x)^2}{-\sin(1)^2 - 2 (e^x)^2 + 2 (e^x)^2 \cos(1) + (e^x)^2 \sin(1)^2} \right) \quad (3.3.2)$$

```
> eval(soln,x=1); evalf(%);
```

$$y(1) = -\arctan \left(\frac{2 e \sin(1) (-1 + \cos(1))}{2 (e)^2 \cos(1) + \sin(1)^2 + (e)^2 \sin(1)^2 - 2 (e)^2} \right) + \pi \\ y(1) = 1.956294970 \quad (3.3.3)$$

d. Comment on the long-term behavior of solutions to this differential equation.

The slope field suggests that as x approaches infinity, all solutions have a horizontal asymptote. That is, as t approaches infinity the solutions have a finite limit (depending on the initial value). The arctan function in the solution formula confirms this observation. One can also appeal to the fact that the values $y = k \pi$ are stationary points on the phase line. All solutions are either attracted to $k \pi$ or repelled from $k \pi$ for some k .

▼ Section 4. Approximate Solutions

7. Modify the ImpEuler procedure to make a procedure that implements the classical Runge-Kutta algorithm (see Simmons/Krantz, Chapter 9, Section 5). Test in by applying it to the problems described in Exercises 2 - 6. (Copy and Paste).

```
> RK := proc(f, x0, y0, h, N)
  local n, Z, X, Y;
  X[0] := x0;
  Y[0] := y0;
  for n from 0 to N-1
  do
    X[n+1] := X[n] + h;
    Z[1] := f(X[n], Y[n]);
    Z[2] := f(X[n] + 0.5*h, Y[n] + 0.5*Z[1]*h);
    Z[3] := f(X[n] + 0.5*h, Y[n] + 0.5*Z[2]*h);
    Z[4] := f(X[n+1], Y[n] + Z[3]*h);
    Y[n+1] := Y[n] + 1/6*(Z[1]+2*Z[2]+2*Z[3]+Z[4])*h;
  end do;
  [ [X[m], Y[m]] $ m=0..N ]
end proc;
```

$RK := \text{proc}(f, x0, y0, h, N)$ (3.4.1)

local n, Z, X, Y ;

$X[0] := x0$;

$Y[0] := y0$;

for n from 0 to $N - 1$ do

$X[n + 1] := X[n] + h$;

$Z[1] := f(X[n], Y[n])$;

$Z[2] := f(X[n] + 0.5 * h, Y[n] + 0.5 * Z[1] * h)$;

$Z[3] := f(X[n] + 0.5 * h,$

$Y[n] + 0.5 * Z[2] * h)$;

$Z[4] := f(X[n + 1], Y[n] + Z[3] * h)$;

$Y[n + 1] := Y[n] + 1/6 * (Z[1] + 2 * Z[2] + 2 * Z[3] + Z[4]) * h$

end do;

[\backslash '([$X[m], Y[m]$], $m=0..N$)]

end proc

Test on Exercise 5.

```
> DE := diff(y(x), x) = cos(x)*y(x);
f := (x, y) -> cos(x)*y;
```

$$DE := \frac{d}{dx} y(x) = \cos(x) y(x)$$


$$f := (x, y) \rightarrow \cos(x) y$$

(3.4.2)


```
> soln := dsolve( {DE, y(0)=1} );
g := unapply(rhs(soln),x);
```

$$\begin{aligned} \text{soln} &:= y(x) = e^{\sin(x)} \\ g &:= \text{exp}@ (x \rightarrow \sin(x)) \end{aligned} \tag{3.4.3}$$

```
> L1 := RK(f,0,1,0.5,20);
plot( [L1,g(x),L1], x=0..10, y=0..3, style=[line$2,point],
color=[red,blue,black]);
```



```
> L2 := RK(f,0,1,0.25,40);
plot( [L2,g(x),L2], x=0..10, y=0..3, style=[line$2,point],
color=[red,blue,black]);
```



```
> Matrix([ [t, 0.5*k $ k=0..5],
['h=0.5', 'L1[k,2]' $ k = 1..6 ],
['h=0.25', 'L2[2*k+1,2]' $ k = 0..5 ],
['g(x)', g(0.5*k) $ k = 0..5 ] ] ): evalf[6](%);
```

t	0.	0.5	1.0	1.5	2.0	2.5
$h=0.5$	1.	1.61486	2.31919	2.71076	2.48190	1.81876
$h=0.25$	1.	1.61513	2.31974	2.71144	2.48254	1.81931
$e^{\sin(x)}$	1.	1.61515	2.31978	2.71148	2.48258	1.81934

(3.4.4)

▼ Part IV. Linear Differential Equations

▼ Section 1. Linear Oscillators

2.* Consider now the following damped system. Obtain the solution.

$$y'' + y' + 4y = 0, y(0) = 2, y'(0) = -3$$

From the form of the solution decide if the system is underdamped, critically damped, or overdamped.

```
> DE := diff(y(t),t,t) + diff(y(t),t) + 4*y(t) = 0;
  inits := y(0)=2, D(y)(0)=-3;
```

$$DE := \frac{d^2}{dt^2} y(t) + \frac{d}{dt} y(t) + 4y(t) = 0$$

$$inits := y(0) = 2, D(y)(0) = -3 \quad (4.1.1)$$

```
> soln := dsolve( {DE,inits} );
```

$$soln := y(t) = -\frac{4}{15} \sqrt{15} e^{-\frac{1}{2}t} \sin\left(\frac{1}{2} \sqrt{15} t\right) + 2 e^{-\frac{1}{2}t} \cos\left(\frac{1}{2} \sqrt{15} t\right) \quad (4.1.2)$$

The system is underdamped.

a. What is the pseudo-period of the oscillations?

The pseudo-period is

```
> 2*Pi/(sqrt(15)/2): evalf[4](%)*time_units;
3.246 time_units \quad (4.1.3)
```

b. What is the time constant?


The time constant is 2 time units.

c. Based upon your answer to part b estimate the time interval required for the oscillations to disappear from view.

The oscillations will disappear in about $5 \times 2 = 10$ time units.

d. Plot the solution curve over the interval you named in part c.

```
> plot( rhs(soln), t=0..10, -2..2,
  tickmarks=[ [2*k$K=0..5], [$-2..2] ] );
```




e. Add to the curve in part d the curves defined by $A e^{-\frac{t}{2}}$ and $A e^{\frac{t}{2}}$, where

$$A = \sqrt{4 + \frac{16}{15}}.$$

Make them blue. What is the significance of these curves? Where did the formula for A come from?

```
> A := sqrt(4 + 16/15):
plot( [rhs(soln), A*exp(-t/2), -A*exp(-t/2)], t=0..10,
      color=[red,blue$2]);
```




The blue curves are the envelopes for the oscillations.

The value of A is derived from the standard formula used to convert the sum of a sine and a cosine into a sinusoid.


f. Use **unapply** to convert the solution into the function g . Use g to plot the phase plane trajectory.

```
> g := unapply(rhs(soln), t):
plot( [g(t), D(g)(t), t=0..10], color=blue );
```




g. Add to the trajectory the points corresponding to $t = 0, 0.25, 0.5, 0.75, \dots, 2.0$.

```
> Points := [g(t/4), D(g)(t/4)] $ t=0..8:
plot( [ [g(t), D(g)(t), t=0..10], [ Points ] ],
      style=[line,point], color=[blue,black] );
```




4.* Use DEplot to draw the direction field in phase space for the damped system. Then add the solution trajectory.

```
> DEsystem := diff(y(t),t) = v(t), diff(v(t),t)=-v(t) - 4*y(t):
> DEtools[DEplot]( {DEsystem}, [y(t),v(t)], t=0..10, y=-2..2,
                   v=-4..2, dirgrid=[11,11], arrows=slim);
```



```
> DEtools[DEplot]( {DEsystem}, [y(t),v(t)], t=0..10, y=-2..2,
                   v=-4..2, dirgrid=[11,11], arrows=slim,
                   {[y(0)=2,v(0)=-3]}, linecolor=blue,
                   stepsize=0.1);
```



▼ Section 2. State Space

2.* Damp the system slightly by changing the IVP to the following

$$y'' + 0.1y' + 4y = \cos(1.8t) \quad , y(0) = 2 \quad , y'(0) = -3$$

Obtain the solution, call it soln and convert it into a function g using unapply. Suppress the output for soln and the definition of g and then enter

```
evalf[3]( 'g(t)' = g(t) )
```

to see a nice looking representation of the solution formula.

```
> DE := diff(y(t),t,t) + 0.1*diff(y(t),t) + 4*y(t) = cos(1.8*t);
inits := y(0)=2, D(y)(0)=-3;
```

$$DE := \frac{d^2}{dt^2} y(t) + 0.1 \left[\frac{d}{dt} y(t) \right] + 4y(t) = \cos(1.8t)$$

$$inits := y(0) = 2, D(y)(0) = -3 \quad (4.2.1)$$

```
> soln := dsolve( {DE,inits} ):
g := unapply(rhs(soln),t):
evalf[3]( 'g(t)'=g(t) );
```

$$g(t) = -1.75 e^{-0.0500t} \sin(2.00t) + 0.754 e^{-0.0500t} \cos(2.00t) + 1.25 \cos(1.80t) + 0.295 \sin(1.80t) \quad (4.2.2)$$

a. Based upon the solution formula determine the time constant for the beats. That is, how long will it take (approximately) for the beats to disappear from the solution curve as it settles down to its steady-state mode?

The time constant is $1/0.05$ time units.


```
> 1/0.05*time_units;
```

$$20.00000000 \text{ time_units} \quad (4.2.3)$$

It will take approximately 100 seconds for the beats to disappear from the solution curve.


b. Plot the solution curve to verify your answer to part a.

```
> plot( g(t), t=0..100, tickmarks = [[20*k$k=0..5],5] );
```




c. Use g to obtain the phase plane and state space trajectories for this system.

```
> plot( [g(t),D(g)(t),t=0..100], title="Phase Plane",  
        color=blue);
```




```
> plots[spacecurve]( [g(t),D(g)(t),t,t=0..100], numpoints=600,  
                    orientation=[-50,70], axes=boxed,  
                    labels=["Position","Velocity","Time"],  
                    color=green, title="State Space" );
```



5.* Apply $DEsystem$ it to the differential equation in Exercise 2 in this section and use $DEplot$ to


obtain the solution trajectory.

```
> DEsystem := diff(y(t),t) = v(t),  
              diff(v(t),t) = -0.1*v(t) - 4*y(t) + cos(1.8*t):  
DEtools[DEplot]( {DEsystem}, [y(t),v(t)], t=0..100, y=-2.2..2.2,  
                  v=-5..4, linecolor=blue, stepsize=0.1,  
                  {[y(0)=2,v(0)=-3]}, thickness=1,  
                  title="Phase Space");
```



6.* Continuing 5. Apply DEplot3d to obtain the state space trajectory for the IVP in Exercise 2.

```
> DEtools[DEplot3d]( {DEsystem}, [y(t),v(t)], t=0..100,  
                     y=-2.5..2.2,  
                     v=-6..5, scene=[y,v,t], orientation=[-50,70],  
                     axes=boxed, labels=["Position","Velocity","Time"],  
                     title="State Space", linecolor=green,  
                     stepsize=0.1, {[y(0)=2,v(0)=-3]}, thickness=1 );
```



▼ Section 3. Two Dimensional Systems

4.* The model

$$x' = x \left(1 - y - \frac{x}{a} \right)$$

$$y' = y \left(1 - x - \frac{y}{b} \right)$$

is used to study competing species. Use **DEplot** to do the following.

- a. Draw the direction field when $1/a = 1.9$ and $1/b = 1.5$. Use the window $\mathbf{x=0..1}$, $\mathbf{y=0..1}$.


> **with(DEtools):**

> **DEsystem := diff(x(t),t) = x(t)*(1 - y(t) - 1.9*x(t)),
diff(y(t),t) = y(t)*(1 - x(t) - 1.5*y(t));**

$$DEsystem := \frac{d}{dt} x(t) = x(t) (1 - y(t) - 1.9x(t)), \quad (4.3.1)$$

$$\frac{d}{dt} y(t) = y(t) (1 - x(t) - 1.5y(t))$$

> **DEplot({DEsystem}, [x(t),y(t)], t=0..10, x=0..1, y=0..1,
arrows=slim, dirgrid=[13,13]);**
DF := %:




- b. Find the stationary point and add the nullclines to the direction field. Discuss the solutions based upon the picture you see.

> **stpt := solve({1-y-1.9*x,1-x-1.5*y},{x,y}):**
StatPoint := subs(stpt,[x,y]);
**Nullclines := plots[implicitplot]([1-y-1.9*x,1-x-1.5*y],
x=0..1, y=0..1, color=green,
thickness=2):**

$$StatPoint := [0.2702702703, 0.4864864865] \quad (4.3.2)$$

> **plots[display](DF, Nullclines,
plot([StatPoint],style=point,color=black));**




Solutions will stream towards the stationary point, moving as indicated in the four regions determined by the nullclines.

c. Add solution curves corresponding to the following set of initial conditions (a circle of points around the stationary point)

$$\{[0, 0.3 + 0.2 \cos(\pi/6 \cdot k), 0.5 + 0.2 \sin(\pi \cdot k/6)] \text{ } \$ \text{ } k=0..11\}$$

```
> DF := DEplot( {DEsystem}, [x(t),y(t)], t=-10..10, x=0..1,
  y=0..1, arrows=slim, dirgrid=[13,13], linecolor=blue,
  thickness=1,
  {[0, 0.3+0.2*cos(Pi/6*k), 0.5+0.2*sin(Pi*k/6)]$k=0..11}):
plots[display]( DF, Nullclines, plot( [StatPoint],
  style=point, color=black, symbol=circle,
  symbolsize=12, thickness=2));
```



▼ Section 4. Matrix Methods

3.* Define the three column vectors b_1 , b_2 , b_3 as follows

$$b_1, b_2, b_3 := \langle 0, 1, -1 \rangle, \langle 1, 1, 0 \rangle, \langle -1, 0, 1 \rangle;$$

```
> with(LinearAlgebra):
```

```
> b1, b2, b3 := <0, 1, -1>, <1, 1, 0>, <-1, 0, 1>;
```

$$b1, b2, b3 := \begin{bmatrix} 0 \\ 1 \\ -1 \end{bmatrix}, \begin{bmatrix} 1 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix} \quad (4.4.1)$$

a. Define the matrix B having b1, b2, b3 as its columns with the entry $\mathbf{B} := \langle \mathbf{b1} | \mathbf{b2} | \mathbf{b3} \rangle$.

> $\mathbf{B} := \langle \mathbf{b1} | \mathbf{b2} | \mathbf{b3} \rangle$;

$$B := \begin{bmatrix} 0 & 1 & -1 \\ 1 & 1 & 0 \\ -1 & 0 & 1 \end{bmatrix} \quad (4.4.2)$$

b. Find the characteristic polynomial of B and factor it with **factor(%)**.

> **CharacteristicPolynomial(B,t);**
factor(%);

$$t^3 - 2t^2 - t + 2 \\ (t-1)(t-2)(t+1) \quad (4.4.3)$$

c. Find B's eigenvalues with **Eigenvalues(B)**.

> **Eigenvalues(B);**

$$\begin{bmatrix} -1 \\ 1 \\ 2 \end{bmatrix} \quad (4.4.4)$$

d. Find B's eigenvectors using **Eigenvectors(B)**. Name the input **lambda,V**.

> **lambda,V := Eigenvectors(B);**

$$\lambda, V := \begin{bmatrix} 2 \\ -1 \\ 1 \end{bmatrix}, \begin{bmatrix} -1 & 2 & 0 \\ -1 & -1 & 1 \\ 1 & 1 & 1 \end{bmatrix} \quad (4.4.5)$$

e. Calculate **MatrixInverse(V).B.V**.

> **MatrixInverse(V).B.V;**

$$\begin{bmatrix} 2 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad (4.4.6)$$

The result is a diagonal matrix showing that B is a diagonalizable matrix. Note that the numbers on the main diagonal are B's eigenvalues.

4.* Continuing 3. Obtain the solution to $\mathbf{v}' = \mathbf{B}\mathbf{v}$ satisfying $\mathbf{v}(0) = \langle 1, 2, 3 \rangle$.

a. Do it first using dsolve applied to the system of linear differential equations defined by $v' = Bv$ with the appropriate initial conditions.

```
> u := <x(t), y(t), z(t)>;
DEsystem := 'diff(u[k], t) = (B.u)[k]' $ k=1..3;
```

$$DEsystem := \frac{d}{dt} x(t) = y(t) - z(t), \frac{d}{dt} y(t) = x(t) + y(t), \frac{d}{dt} z(t) = -x(t) + z(t) \quad (4.4.7)$$

```
> soln := dsolve( {DEsystem, x(0)=1, y(0)=2, z(0)=3} );
```

$$soln := \left\{ z(t) = \frac{1}{2} e^{-t} + \frac{5}{2} e^t, x(t) = e^{-t}, y(t) = -\frac{1}{2} e^{-t} + \frac{5}{2} e^t \right\} \quad (4.4.8)$$

```
> VectorForm = subs(soln, u);
```

$$VectorForm = \begin{bmatrix} e^{-t} \\ -\frac{1}{2} e^{-t} + \frac{5}{2} e^t \\ \frac{1}{2} e^{-t} + \frac{5}{2} e^t \end{bmatrix} \quad (4.4.9)$$

```
> Check;
eval(%%, t=0);
eval(DEsystem, soln);
```

Check

$$VectorForm = \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}$$

$$-e^{-t} = -e^{-t}, \frac{1}{2} e^{-t} + \frac{5}{2} e^t = \frac{1}{2} e^{-t} + \frac{5}{2} e^t, -\frac{1}{2} e^{-t} + \frac{5}{2} e^t = -\frac{1}{2} e^{-t} + \frac{5}{2} e^t \quad (4.4.10)$$

b. Do it second by making a fundamental matrix solution $X(t)$ defined as the matrix with the eigenvector solutions in the columns then computing

$$v(t) = X(t) X(0)^{-1} v(0)$$

```
> for k from 1 to 3
do
W[k] := exp(lambda[k]*t)*Column(V, k);
end do;
unassign('k');
```

```
> <W[1]|W[2]|W[3]>;
X := unapply(%, t):
```

$$X = \begin{bmatrix} -e^{2t} & 2e^{-t} & 0 \\ -e^{2t} & -e^{-t} & e^t \\ e^{2t} & e^{-t} & e^t \end{bmatrix} \quad (4.4.11)$$

> `v := X(t).X(0)^(-1).<1,2,3>;`

$$v := \begin{bmatrix} e^{-t} \\ -\frac{1}{2}e^{-t} + \frac{5}{2}e^t \\ \frac{1}{2}e^{-t} + \frac{5}{2}e^t \end{bmatrix} \quad (4.4.12)$$

> `Check;`
`eval(v,t=0);`
`map(diff,v,t) = B.v;`

Check

$$\begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix} \begin{bmatrix} -e^{-t} \\ \frac{1}{2}e^{-t} + \frac{5}{2}e^t \\ -\frac{1}{2}e^{-t} + \frac{5}{2}e^t \end{bmatrix} = \begin{bmatrix} -e^{-t} \\ \frac{1}{2}e^{-t} + \frac{5}{2}e^t \\ -\frac{1}{2}e^{-t} + \frac{5}{2}e^t \end{bmatrix} \quad (4.4.13)$$

c. To it third by using the matrix exponential, Exp_At . Once you have it, the solution is

$$v(t) = Exp_At v(0)$$

> `Exp_At := MatrixExponential(B,t);`

$$Exp_At := \begin{bmatrix} \frac{1}{3}e^{2t} + \frac{2}{3}e^{-t} & \frac{1}{3}e^{2t} - \frac{1}{3}e^{-t} & -\frac{1}{3}e^{2t} + \frac{1}{3}e^{-t} \\ \frac{1}{3}e^{2t} - \frac{1}{3}e^{-t} & \frac{1}{3}e^{2t} + \frac{1}{6}e^{-t} + \frac{1}{2}e^t & -\frac{1}{3}e^{2t} - \frac{1}{6}e^{-t} + \frac{1}{2}e^t \\ -\frac{1}{3}e^{2t} + \frac{1}{3}e^{-t} & -\frac{1}{3}e^{2t} - \frac{1}{6}e^{-t} + \frac{1}{2}e^t & \frac{1}{3}e^{2t} + \frac{1}{6}e^{-t} + \frac{1}{2}e^t \end{bmatrix} \quad (4.4.14)$$

> `v := Exp_At.<1,2,3>;`

$$v := \begin{bmatrix} e^{-t} \\ -\frac{1}{2}e^{-t} + \frac{5}{2}e^t \\ \frac{1}{2}e^{-t} + \frac{5}{2}e^t \end{bmatrix} \quad (4.4.15)$$

> `Check;`

See_above;

Check

See_above

(4.4.16)

d. Check the solution in each case.

▼ Section 5. The Laplace Transform

6.* Consider the following initial value problem

$$y'' + y' + 9y = H(t - 2\pi) - H(t - 5\pi) + 2\text{Dirac}(t - 9\pi), \quad y(0) = 0, y'(0) = 1$$

a. Obtain the solution using dsolve. Plot it for $t = 0..40$ and explain the behavior of the solution curve.

> alias(H=Heaviside,delta=Dirac);

H, δ

(4.5.1)

> DE := diff(y(t),t,t) + diff(y(t),t) + 9*y(t) =
H(t-2*Pi)-H(t-5*Pi)+2*delta(t-9*Pi);
inits := y(0)=0, D(y)(0)=1;

$$DE := \frac{d^2}{dt^2} y(t) + \frac{d}{dt} y(t) + 9y(t) = H(t - 2\pi) - H(t - 5\pi) + 2\delta(t - 9\pi)$$


$$inits := y(0) = 0, D(y)(0) = 1$$

(4.5.2)

> soln := dsolve({DE,inits}, y(t));


$$soln := y(t) = \frac{2}{35} e^{-\frac{1}{2}t} \sin\left(\frac{1}{2}\sqrt{35}t\right) \sqrt{35} - \frac{1}{9} H(t - 2\pi) e^{-\frac{1}{2}t + \pi} \cos\left(\frac{1}{2}\sqrt{35}(-t + 2\pi)\right) + \frac{1}{315} \sqrt{35} H(t - 2\pi) e^{-\frac{1}{2}t + \pi} \sin\left(\frac{1}{2}\sqrt{35}(-t + 2\pi)\right) + \frac{1}{9} H(t - 2\pi) - \frac{1}{9} H(t - 5\pi) - \frac{1}{315} \sqrt{35} H(t - 5\pi) e^{-\frac{1}{2}t + \frac{5}{2}\pi} \sin\left(\frac{1}{2}\sqrt{35}(-t + 5\pi)\right) + \frac{1}{9} H(t - 5\pi) e^{-\frac{1}{2}t + \frac{5}{2}\pi} \cos\left(\frac{1}{2}\sqrt{35}(-t + 5\pi)\right) - \frac{4}{35} \sqrt{35} H(t - 9\pi) e^{-\frac{1}{2}t + \frac{9}{2}\pi} \sin\left(\frac{1}{2}\sqrt{35}(-t + 9\pi)\right)$$

> plot(rhs(soln), t=0..40);



The solution curve shows the response to the steady force from $t = 6$ to $t = 16$ (roughly). See the plot of the driver below. Then there is no force until the kick provided by the Dirac delta at about $t = 27$. Note that at $t = 27$ the system reacts as it did at $t = 0$ except it goes twice as far from equilibrium.

```
> plot( rhs(DE), t=0..40, title="Heaviside in Driver");
```



b. Obtain the solution using `dsolve/method=laplace`. How does the solution formula compare to the formula obtained in part a?

```
> soln2 := dsolve( {DE,inits}, y(t), method=laplace );
```

$$\begin{aligned}
 \text{soln2} := y(t) &= \frac{2}{35} e^{-\frac{1}{2}t} \sin\left(\frac{1}{2} \sqrt{35} t\right) \sqrt{35} & (4.5.4) \\
 &+ \frac{4}{35} \sqrt{35} H(t-9\pi) e^{-\frac{1}{2}t + \frac{9}{2}\pi} \sin\left(\frac{1}{2} \sqrt{35} (t-9\pi)\right) + \frac{1}{630} \left(70 \right. \\
 &+ \text{Ie}^{\frac{1}{2}\text{I}(1+\sqrt{35})(t-2\pi)} (35\text{I} + \sqrt{35}) - \text{Ie}^{-\frac{1}{2}\text{I}(t-2\pi)(-1+\sqrt{35})} (-35\text{I} \\
 &+ \sqrt{35}) \left. \right) H(t-2\pi) + \frac{1}{630} \left(-70 + \text{Ie}^{-\frac{1}{2}\text{I}(t-5\pi)(-1+\sqrt{35})} (-35\text{I} \right. \\
 &+ \sqrt{35}) - \text{Ie}^{\frac{1}{2}\text{I}(1+\sqrt{35})(t-5\pi)} (35\text{I} + \sqrt{35}) \left. \right) H(t-5\pi)
 \end{aligned}$$

The Laplace solution is in complex form, probably because of a factorization of the characteristic polynomial over the complex field.


7.* Use `unapply` to convert the solution to 6 a into a function, `g`. Use `g` to plot the phase plane trajectory and as well as the state space trajectory for the system. Use the following options in the

spacecurve procedure for the state space trajectory


```
axes=boxed, orientation=[-60,70], color=red, numpoints=600,  
labels=["Position", "Velocity", "Time"]
```

```
> g := unapply(rhs(soln), t):
```

```
> plot( [g(t), D(g)(t), t=0..100], title="Phase Space Trajectory",  
color=blue);
```



```
> plots[spacecurve]( [g(t), D(g)(t), t, t=0..100],  
axes=boxed, orientation=[-50,70],  
color=green, numpoints=800,  
labels=["Position", "Velocity", "Time"] );
```



8.* Continuing 7. Use the entry

```
plots[animate]( plot, [ [g(t), D(g)(t), t=0..T] ], T=0..40, frames=82);
```

to plot an animation of the phase space trajectory. Click on the plot and use the video controls in the context bar to play the animation and also to view it one frame at a time. Compare the trajectory movement to the time series for position.

Note. The slide control can also be used to control the display of the frames. Assuming t is measured in seconds, reduce the frame rate to 2 per second to view the trajectory in "real time".

```
> plots[animate]( plot, [ [g(t), D(g)(t), t=0..T], color=blue ],  
T=0..40, frames=82);
```

$$T = .49383$$

